

ЧАСТЬ II

# Встроенные функции Excel и их использование

**В этой части...**

- Глава 6. Введение в функции Excel
- Глава 7. Логические, текстовые и информационные функции
- Глава 8. Работа с числовыми данными
- Глава 9. Функции для работы с массивами значений и списками (базами данных)
- Глава 10. Статистические и инженерные функции
- Глава 11. Финансовые функции
- Глава 12. Массивы констант и формулы массива
- Глава 13. Таблицы, сводные таблицы и аналитические функции



# Введение в функции Excel

---

## *В этой главе...*

- Понятие функции Excel
  - Категории встроенных функций
  - Общие правила работы с функциями
  - Функции СУММ, СРЗНАЧ, МИН, МАКС, СЧЁТ и СЧЁТЗ
- 

В предыдущих главах мы описали базовые понятия и основные приемы работы в Excel, которые должен знать каждый пользователь. В этой главе мы рассмотрим, что представляет собой встроенная функция Excel, а также познакомимся с некоторыми общими правилами работы с функциями.

При составлении несложных формул можно обойтись без функций. Действительно, зачем использовать функцию СУММ там, где требуется сложить два-три числа? Однако там, где требуется сложить 10 чисел и больше, вместо длинной формулы

$$=A1+A2+A3+A4+A5+A6+A7+A8+A9+A10$$

лучше использовать эквивалентную ей формулу `=СУММ(A1:A10)` с функцией СУММ. Использование функции СУММ вместо длинного математического выражения сделало формулу более компактной и понятной.

В большинстве случаев использование функций в формулах позволяет не только упростить саму формулу, но и выполнять такие вычисления, которые осуществить по-другому, без функций, просто невозможно. Предположим, требуется найти наибольшее и наименьшее число в диапазоне A1:A10. Без функций МИН и МАКС, которые возвращают максимальное и минимальное значения из набора данных, здесь не обойтись. Формулы `=МАКС(A1:A10)` и `=МИН(A1:A10)` вычисляют максимальное и минимальное значения в заданном диапазоне A1:A10.

С помощью функций можно выполнять редактирование данных в ячейках и поиск конкретных значений в электронных таблицах. Например, функция ПРОПНАЧ заменяет текст в ячейке строчными буквами, оставляя прописной только первую букву слова, а с помощью таких встроенных функций, как ПРОСМОТР, ВПР и ГПР, можно успешно осуществлять поиск данных в электронных таблицах.

И наконец, можно сделать свои формулы более гибкими, добавив в них функции, которые выполняют вычисления в соответствии с заданными условиями. Вспомним, например, как использование функции ЕСЛИ при вычислении выражения  $1/x$  в главе 5 позволило избежать появления ошибки #ДЕЛ/0! для  $x = 0$ .

## Понятие функции Excel

По сути, каждая встроенная функция в Excel — это заранее определенная формула (или несколько формул). Каждая встроенная функция Excel имеет название или *уникальное имя*. В формулах функция выглядит следующим образом: **Название функции (Аргументы)**, где **Название функции** — это уникальное имя функции, а **Аргументы** — это список *аргументов* функции или, другими словами, ее *входные данные*.

Встроенная функция Excel работает по принципу “черного ящика”. Вы задаете функции входные данные или аргументы, а функция вычисляет некоторый *результат* или возвращает вам *выходные данные*. При работе с функциями главная задача пользователя — правильно задать ее аргументы. Непосредственно процесс получения результата скрыт от глаз пользователя. Мы не видим, по каким именно формулам проводятся вычисления. В ячейке отображается лишь результат, возвращаемый функцией. Например, в формулах `=МИН(A1:A10;C1:C10)`, `=СУММ(A1:A10;C1:C10)` аргументами или входными данными функций МИН и СУММ являются ссылки на диапазоны A1:A10 и C1:C10. Выходными данными функции МИН является минимальное числовое значение в исходном диапазоне, а выходными данными функции СУММ — сумма всех чисел в заданном диапазоне.

Список аргументов функции может быть пуст, т.е. функция может не иметь аргументов. В качестве примера таких функций можно привести функцию ПИ (формула `=ПИ()`), которая возвращает число  $\pi$  с точностью до 15 цифр (рис. 6.1), функцию СЕГОДНЯ (формула `=СЕГОДНЯ()`), которая возвращает текущую дату, или функцию СЛЧИС (формула `=СЛЧИС()`), которая возвращает случайное число в диапазоне от 0 до 1.



Обратите внимание на то, что даже если у функции нет аргументов, наличие скобок после имени функции обязательно. В противном случае Excel не распознает функцию и возвращает значение ошибки #ИМЯ? (рис. 6.2).

	A	B	C	D
1		Результат		Формула
2		15.03.2010		=СЕГОДНЯ()
3		3,141592654		=ПИ()
4		0,408053031		=СЛЧИС()
5				
6				
7	17.05.2006		1398	=СЕГОДНЯ()-A7
8		1,570796327		=ПИ()/2
9		0,443098686		=1-СЛЧИС()
10	0,408053			=1-A10
11				

Рис. 6.1. Примеры функций без аргументов

	A	B	C	D
14				
15		#ИМЯ?		=слчис
16		#ИМЯ?		=пи
17		#ИМЯ?		=сегодня
18				
19				Формула содержит нераспознанный текст.
20				

Рис. 6.2. Excel не распознает функции ПИ, СЛЧИС и СЕГОДНЯ, если после имен этих функций не указаны скобки

Многие функции Excel имеют один аргумент. Например, один аргумент имеют следующие функции.

- ▶ Математические функции SIN, COS и TAN (формулы `=SIN(число)`, `=COS(число)` и `=TAN(число)`). Они вычисляют синус, косинус и тангенс угла (рис. 6.3). Аргумент этих функций **число** есть величина угла, выраженная в радианах. Функция

РАДИАНЫ преобразует градусы в радианы. Эта функция также имеет один аргумент — **угол**, который представляет собой величину угла, выраженную в градусах.

- ▶ Текстовые функции КОДСИМВ и СИМВОЛ (формулы =КОДСИМВ(текст) и =СИМВОЛ(число)). Эти функции возвращают цифровой код для первого символа текстовой строки или, наоборот, символ, которому соответствует заданный цифровой код (см. рис. 6.3). Аргумент **текст** функции КОДСИМВ — текстовая строка, аргумент **число** функции СИМВОЛ — это цифровой код символа.
- ▶ Функции для работы с датой и временем ГОД, МЕСЯЦ и ДЕНЬ (формулы =ГОД(дата\_в\_числовом\_формате), =МЕСЯЦ(дата\_в\_числовом\_формате) и =ДЕНЬ(дата\_в\_числовом\_формате)). Они извлекают из своего аргумента — **дата\_в\_числовом\_формате** — год, месяц и день соответственно (см. рис. 6.3).
- ▶ Другие функции. Например, функция ЕОШИБКА (формула =ЕОШИБКА(значение)) возвращает логическое значение ИСТИНА, если ее аргумент **значение** оказывается одним из значений ошибки #Н/Д, #ИМЯ? и т.п. (см. рис. 6.3).

Если функция имеет больше одного аргумента, то все ее аргументы разделяются точкой с запятой. Порядок следования аргументов в некоторых случаях не имеет значения, в некоторых — принципиально важен. На рис. 6.4 приведены формулы, содержащие функции ДАТА, ВРЕМЯ, БС, ПС и ПРАВСИМВ (ячейки D2:D6 и ячейка D10). Они возвращают правильный ре-

зультат только в том случае, если их аргументы заданы в правильном порядке. Для функций СУММ, ПРОИЗВЕД или И (ячейки D8:D9 и B12:B15) порядок следования аргументов не принципиален. Вы можете задавать их в произвольном порядке. Функция ЕСЛИ содержит всего три аргумента, но порядок следования ее аргументов строго определен. Первый аргумент — это логическое выражение, второй аргумент — константа, ссылка или формула, которая вычисляется в том случае, если логическое выражение истинно, и, наконец, третий аргумент — константа, ссылка или формула, которая вычисляется в том случае, если логическое выражение ложно. Попробуйте переставить второй и третий аргументы местами, и вы получите формулу, работающую с точностью до наоборот.

Функции могут иметь как строго фиксированное, так и неопределенное количество аргументов. Например, функции И, СУММ и ПРОИЗВЕД могут иметь до 255 аргументов, разделенных точкой с запятой (рис. 6.5). При вводе функции в ячейку на экране появляется подсказка, содержащая список всех аргументов функции; если функция содержит неопределенное количество аргументов, то в конце этого списка отображается троеточие.

	A	B	C	D	E
1	Данные	Результат	Формула		
2	1,04719755	0,866025404	=SIN(A2)		
3	1,04719755	0,5	=COS(A3)		
4	1,04719755	1,732050808	=TAN(A4)		
5					
6	200	50	=КОДСИМВ(A6)		
7	200	и	=СИМВОЛ(A7)		
8					
9	23.02.2010	2010	=ГОД(A9)		
10	23.02.2010	2	=МЕСЯЦ(A10)		
11	23.02.2010	23	=ДЕНЬ(A11)		
12					
13	12	ложь	=ЕОШИБКА(A13)		
14	#Н/Д	ИСТИНА	=ЕОШИБКА(A14)		
15					

Рис. 6.3. Примеры функций с одним аргументом



В данной книге в качестве символа-разделителя аргументов функций используется точка с запятой.

	A	B	C	D	E	F	G	H	I	
1		Данные		Результат		Формула				
2		2010	12	15	15.12.2010	ДАТА(A2;B2;C2)				
3		20	20	12	20:20:12	=ВРЕМЯ(A3;B3;C3)				
4		0,15	12	300	80 253,75р.	=BC(A4;B4;)-A5)				
5		15000	25	25	-2 803,61р.	=ПС(A4;B4;A5)				
6					10	=РАНГ.РВ(C4;A2;C5;1)				
7					\$A\$5	=ЕСЛИ(A2>A5;ЯЧЕЙКА("адрес";A2);ЯЧЕЙКА("адрес";A5))				
8					ИСТИНА	=И(B2<C2;A3<C4;C3<C4)				
9					ЛОЖЬ	=И(B2>C2;A3>C4;C3>C4)				
10					кот	=ПРАВСИМВ("антрекот";3)				
11	Порядок следования аргументов функций СУММ и ПРОИЗВЕД не имеет значения									
12					421	=СУММ(B2;B5;C2;C5)				
13					421	=СУММ(C2;C5;B2;B5)				
14					9,7E+10	=ПРОИЗВЕД(B2;B5;C2;C5)				
15					9,7E+10	=ПРОИЗВЕД(C2;C5;B2;B5)				
16										

Рис. 6.4. Примеры функций с несколькими аргументами

При вводе функции появляется всплывающая подсказка

	A	B	C	D	E	F	G	H	I	J	K	L
1		Данные		Результат		Формула						
2		1	15	20	292	=СУММ(A2;A7;B2;B7;C2;C7)						
3		2	10	12	292	=СУММ(C2;C7;A2;A7;B2;B7)						
4		3	14	25	75600000	=ПРОИЗВЕД(A2;C2;A3;C3;A4;C4)						
5		4	15	45	75600000	=ПРОИЗВЕД(A4;C4;A2;C2;A3;B3;C3)						
6		5	13	36	1512000	=ПРОИЗВЕД(A2;A3;A4;A5;A6;A7;B2;B3;B4)						
7		6	18	48	20,0625	=СРЗНАЧ(A2;A7;B2;B7;C2;C4;C5;125)						
8					20,0625	=СРЗНАЧ(B2;B7;A2;A7;C2;C4;C5;125)						
9												
10	=произвед(A2;A3;A4;A5;A6;A7;B2;B3;B4;											
11	ПРОИЗВЕД(число1; [число2]; [число3]; [число4]; [число5]; [число6]; [число7]; [число8]; [число9]; [число10]; [число11]; ...)											
12												

Название функции                      Список аргументов                      Вводимый аргумент

Рис. 6.5. Примеры функций с неопределенным количеством аргументов



Символ-разделитель в списке аргументов функций необязательно должен быть точкой с запятой. Можно использовать другой символ, например запятую. Символ-разделитель определяется в поле со списком Разделитель элементов списка, которое расположено на вкладке Число диалогового окна Настройка региональных параметров. Об этом диалоговом окне уже шла речь в главе 4. Напомним, для того чтобы открыть это окно в Windows XP или Windows 2000–2003, необходимо выбрать команду Пуск⇒Панель управления⇒Язык и региональные стандарты и щелкнуть на кнопке Настройка.

Многие функции Excel имеют аргументы, которые задавать не обязательно. Например, функция ПРАВСИМВ возвращает заданное количество последних символов текстовой строки. Формула =ПРАВСИМВ("Антрекот";3) возвращает текст "кот", формула =ПРАВСИМВ("Антрекот";2) — текст "от". Данная функция имеет два аргумента — текст и число\_знаков. Второй аргумент — число\_знаков — задавать не обязательно. Если этот аргумент опущен, то функция возвращает крайний правый символ текстовой строки. Формула =ПРАВСИМВ("Антрекот") возвращает символ "т", также как и формула =ПРАВСИМВ("Антрекот";1). Однако формула =ПРАВСИМВ("Антрекот");) вернет пустую ячейку, так как в этом случае подразумевается, что аргумент число\_

знаков равен нулю. Формула =ПРАВСИМВ("Антрекот";0) также вернет пустую строку. Обратите внимание: если необязательный аргумент этой функции опущен, то его не нужно включать в список аргументов функции.

Рассмотрим еще один пример. Функция ЕСЛИ содержит три аргумента, последние два из которых можно не задавать. Однако, если какой-либо из последних двух аргументов функции ЕСЛИ опущен, в списке аргументов он должен быть отделен точкой с запятой, как показано на рис. 6.6 в ячейках G2:G4.

## Типы аргументов функций

В качестве аргументов функций можно использовать константы, ссылки на ячейки и диапазоны ячеек, имена, математические выражения, массивы значений, а также другие функции.

### Константы и ссылки на ячейки и диапазоны

Аргумент функции может быть константой, т.е. текстовой строкой, числовым или логическим значением. Если функция содержит один аргумент, то использование константы в качестве аргумента не имеет особого смысла, поскольку формула всегда возвращает одно и то же значение. Например, формула =КОРЕНЬ(25) всегда возвращает одно и то же значение — 5. В подобных случаях аргумент функции лучше задать в виде ссылки на ячейку. В нашем примере число 25 можно ввести в одну из ячеек, например A1. В другую ячейку, например A2, можно ввести формулу =КОРЕНЬ(A1) (вместо формулы =КОРЕНЬ(25)). Теперь при каждом изменении значения в ячейке A1 формула в ячейке A2 вычисляет другой результат.

Если функция имеет два и более аргументов, использование констант в качестве аргументов оправдано. Рассмотрим, например, функцию ПОВТОР. Эта функция повторяет текст заданное количество раз. Так, если в ячейке A1 содержится символ "|", то формула =ПОВТОР(A1;10), введенная в ячейку B1, отобразит этот символ 10 раз (рис. 6.7). В данном примере второй аргумент функции ПОВТОР, задающий количество повторений содержимого ячейки A1, является константой.

	A	B	C	D	E	F	G
1	Данные			Результат		Формула	
2	1	15	20			0	=ЕСЛИ(A2>10;5;)
3	2	10	12			15	=ЕСЛИ(A2>10;;15)
4	3	14	25			0	=ЕСЛИ(A2>10;;)
5	4	15	45		ЛОЖЬ		=ЕСЛИ(A2>10;;)
6	5	13	36			0	=ЕСЛИ(A2<10;)
7	6	18	48				

Рис. 6.6. Примеры использования функции ЕСЛИ

	A	B	C	D
1				
2				
3				

Рис. 6.7. Пример использования ссылок и констант в качестве аргументов функций

В качестве аргументов многих функций можно использовать ссылки на диапазон ячеек. Например, в качестве аргументов функций СУММ, СРЗНАЧ, МИН и МАКС в формулах =СУММ(A2:C7), =СРЗНАЧ(A2:C7), =МИН(A2:C7) и =МАКС(A2:C7) используется ссылка на диапазон ячеек A2:C7. При изменении диапазона программа автоматически скорректирует во всех этих формулах ссылку на диапазон. Например, если вы добавите в диапазон A2:C7 новую строку, формулы изменятся следующим образом: =СУММ(A2:C8), =СРЗНАЧ(A2:C8), =МИН(A2:C8) и =МАКС(A2:C8) (рис. 6.8).

	A	B	C	D	E	F	G	H
1		Данные			Результат		Формула	
2	1	15	20		292		=СУММ(A2:C7)	
3	2	10	12		16,22		=СРЗНАЧ(A2:C7)	
4	3	14	25		48		=МАКС(A2:C7)	
5	4	15	45		1		=МИН(A2:C7)	
6	5	13	36					
7	6	18	48					

	A	B	C	D	E	F	G	H
1		Данные			Результат		Формула	
2	1	15	20		292		=СУММ(A2:C8)	
3	2	10	12		16,22		=СРЗНАЧ(A2:C8)	
4	3	14	25		48		=МАКС(A2:C8)	
5	4	15	45		1		=МИН(A2:C8)	
6								
7	5	13	36					
8	6	18	48					

Рис. 6.8. Пример автоматического изменения ссылок на диапазон ячеек

	A	B	C	D	E	F	G
1					Результат		Формула
2	1				105		=СУММ(A:A)
3	2				7,5		=СРЗНАЧ(A:A)
4	3				14		=МАКС(A:A)
5	4				1		=МИН(A:A)

Рис. 6.9. Пример использования ссылки на весь столбец A в качестве аргумента функций СУММ, СРЗНАЧ, МИН и МАКС

В некоторых случаях в качестве аргумента функции удобно использовать ссылку на весь столбец или строку. Например, формулы `=СУММ(A:A)`, `=СРЗНАЧ(A:A)`, `=МИН(A:A)` и `=МАКС(A:A)` вычисляют сумму, среднее значение, минимальное и максимальное значения во всем столбце A (рис. 6.9). Обратите внимание: Excel не использует в вычислениях те ячейки столбца A, которые не содержат значений. Однако при добавлении в этот столбец новых значений формулы автоматически пересчитываются.



Использование ссылки на весь столбец (или строку) особенно эффективно в тех случаях, когда диапазон, ссылка на который используется в качестве аргумента функции, постоянно меняется.

## Использование имен

В качестве аргументов функции можно использовать имена. Если диапазону A2:C7 на рис. 6.8 присвоить имя **Данные**, то вместо ссылок на диапазон в формулах `=СУММ(A2:C7)`, `=СРЗНАЧ(A2:C7)`, `=МИН(A2:C7)` и `=МАКС(A2:C7)` можно использовать имя этого диапазона: `=СУММ(Данные)`, `=СРЗНАЧ(Данные)`, `=МИН(Данные)` и `=МАКС(Данные)`. Еще один пример, приведенный на рис. 6.10, наглядно иллюстрирует преимущества использования имен вместо ссылок. Очевидно, что в таком виде,



как на рисунке, формулы становятся более понятными, несмотря на громоздкий вид некоторых из них (например, формула в ячейке E20).

E20							fx	=СУММ(Микроавтобусы_Альянс_4_кв;Микроавтобусы_Автосалон_4_кв;Микроавтобусы_ЛВ_Авто_4_кв)
A	B	C	D	E	F	G		
	Название магазина	Легковые автомобили	Грузовые автомобили	Микроавтобусы	Всего	Формулы		
4								
5	Альянс	720 000	453 200	890 000	2 063 200	=СУММ(Альянс_1_кв)		
6	Автосалон	220 600	215 000	325 000	760 600	=СУММ(Автосалон_1_кв)		
7	ЛВ-Авто	379 100	798 200	650 000	1 827 300	=СУММ(ЛВ_Авто_1_кв)		
8	Итого	1 319 700	1 466 400	1 865 000	4 651 100	=СУММ(ЛВ_Авто_1_кв;Автосалон_1_кв;Альянс_1_кв)		
9	Альянс	420 000	400 500	620 200	1 440 700	=СУММ(Альянс_2_кв)		
10	Автосалон	280 900	495 200	650 200	1 426 300	=СУММ(Автосалон_2_кв)		
11	ЛВ-Авто	340 200	560 100	360 700	1 261 000	=СУММ(ЛВ_Авто_2_кв)		
12	Итого	1 041 100	1 455 800	1 631 100	4 128 000	=СУММ(Альянс_2_кв;Автосалон_2_кв;ЛВ_Авто_2_кв)		
13	Альянс	290 300	345 500	560 000	1 195 800	=СУММ(Альянс_3_кв)		
14	Автосалон	512 300	548 200	445 200	1 505 700	=СУММ(Автосалон_3_кв)		
15	ЛВ-Авто	500 200	450 100	350 900	1 301 200	=СУММ(ЛВ_Авто_3_кв)		
16	Итого	1 302 800	1 343 800	1 356 100	4 002 700	=СУММ(Альянс_3_кв;Автосалон_3_кв;ЛВ_Авто_3_кв)		
17	Альянс	350 800	325 400	810 700	1 486 900	=СУММ(Альянс_4_кв)		
18	Автосалон	780 200	789 500	405 100	1 974 800	=СУММ(Автосалон_4_кв)		
19	ЛВ-Авто	609 500	250 300	412 300	1 272 100	=СУММ(ЛВ_Авто_4_кв)		
20	Итого	1 740 500	1 365 200	1 628 100	4 733 800	=СУММ(Альянс_4_кв;Автосалон_4_кв;ЛВ_Авто_4_кв)		
21	Альянс	1 781 100	1 524 600	2 880 900	6 186 600	=СУММ(Альянс_год)		
22	Автосалон	1 794 000	2 047 900	1 825 500	5 667 400	=СУММ(Автосалон_год)		
23	ЛВ-Авто	1 829 000	2 058 700	1 773 900	5 661 600	=СУММ(ЛВ_Авто_год)		
24	Итого	5 404 100	5 631 200	6 480 300	17 515 600	=СУММ(Альянс_год;Автосалон_год;ЛВ_Авто_год)		
25				=СУММ(Микроавтобусы)				
26			=СУММ(Грузовые_автомобили)					
27		=СУММ(Легковые_автомобили)						

Рис. 6.10. Примеры использования имен в качестве аргументов функций

## Использование выражений и функций

В качестве аргумента функции можно использовать математические выражения. Например, в формуле =КОРЕНЬ(A5) используется ссылка на ячейку A5. Если эта ячейка содержит формулу =A2\*A3-A4, то ее можно использовать в качестве аргумента функции КОРЕНЬ. В результате получим формулу =КОРЕНЬ(A2\*A3-A4). Если в качестве аргумента функции используется математическое выражение, то Excel сначала вычисляет значение выражения, а затем использует полученный результат в качестве значения аргумента.



Напоминаем, что посмотреть пошаговое вычисление любой формулы можно в диалоговом окне Вычисление формулы (команда Формулы⇒Зависимости формул⇒Вычислить формулу).

Математические выражения, используемые в качестве аргументов функции, в свою очередь, могут содержать другие функции. Например, в формуле =КОРЕНЬ(ПРОИЗВЕД(A2;A3)-A4) в качестве аргумента функции КОРЕНЬ используется математическое выражение, которое содержит функцию ПРОИЗВЕД, вычисляющую произведение своих аргументов; в качестве аргументов функции ПРОИЗВЕД используются ссылки на ячейки A2 и A3.

Функции, которые используются внутри других функций, называются *вложенными*. В рассматриваемом примере с формулой =КОРЕНЬ(ПРОИЗВЕД(A2;A3)-A4) вложенной является функция ПРОИЗВЕД. При вычислении формулы, содержащей вложенные функции, Excel сначала вычисляет вложенную функцию. Если функция

содержит несколько уровней вложенных функций, первым вычисляется выражение с наибольшей глубиной вложения.

В Excel 2010 формула может содержать до 64 уровней вложенных функций. Если формула содержит больше 64 уровней вложенных функций, Excel выдаст сообщение об ошибке. В большинстве случаев использование уже больше пяти–семи уровней вложенных функций не имеет смысла. А если по смыслу решаемой задачи требуется использовать большее число уровней вложенных функций, то необходимо корректировать алгоритм решения.

Наиболее распространенный вариант формулы с несколькими уровнями вложенных функций – это формулы, в которых используются вложенные функции ЕСЛИ. Например, следующая формула:

```
=ЕСЛИ(A1=1;ИСТИНА;ЕСЛИ(A1=3;ИСТИНА;ЕСЛИ(A1=5;ИСТИНА;  
ЕСЛИ(A1=7;ИСТИНА;ЕСЛИ(A1=9;ИСТИНА;ЕСЛИ(A1=11;ИСТИНА;ЛОЖЬ))))))
```

возвращает логическое значение ИСТИНА только тогда, когда в ячейке A1 содержится одно из чисел – 1, 3, 5, 7, 9 или 11. Во всех остальных случаях данная формула возвращает логическое значение ЛОЖЬ. Записать данную формулу можно более компактно, используя массив констант и логическую функцию ИЛИ.

### Использование массивов констант

В Excel в качестве аргумента функции можно использовать массив констант. При использовании массива констант в качестве аргумента функции элементы массива должны быть разделены точкой с запятой или двоеточием и заключены в фигурные скобки. Формула =ИЛИ(A1={1;3;5;7;9;11}) вычисляет значение ИСТИНА, если в ячейке A1 содержится одно из значений – 1, 3, 5, 7, 9 или 11. Тот же самый результат вычисляет формула

```
=ЕСЛИ(ИЛИ(A1={1;3;5;7;9;11});ИСТИНА();ЛОЖЬ())
```

или формула

```
=ЕСЛИ(A1=1;ИСТИНА;ЕСЛИ(A1=3;ИСТИНА;ЕСЛИ(A1=5;ИСТИНА;  
ЕСЛИ(A1=7;ИСТИНА;ЕСЛИ(A1=9;ИСТИНА;ЕСЛИ(A1=11;ИСТИНА;ЛОЖЬ))))))
```

## Категории встроенных функций

Все встроенные функции Excel разделены на несколько категорий. Например, функции категории **Текстовые** используются, в основном, для работы с текстовыми строками. Математические функции, с помощью которых можно составлять различные математические выражения, отнесены к категории **Математические** и т.п.

Основные категории функций перечислены ниже. Все названия категорий соответствуют названиям команд, расположенным в группе **Библиотека функций** на вкладке **Формулы**. Полный список всех встроенных функций Excel приводится в приложении А.

### Логические функции

Категория **Логические** содержит семь функций, в том числе функции ЕСЛИ и ЕСЛИОШИБКА. Использование логических функций делает формулы более гибкими, а использование функции ЕСЛИ наделяет формулу способностью “прини-

мать решения”. Благодаря этому функция ЕСЛИ стала самой используемой логической функцией.



Функция ЕСЛИОШИБКА имеется в библиотеке встроенных функций только в Excel 2010 (2007). Об этом необходимо помнить, если ваши рабочие книги используются в разных версиях Excel.

## Текстовые функции

Текстовые функции предназначены для обработки текста. Например, с помощью функций ПРОПНАЧ или ДЛСТР можно изменить регистр или определить длину текстовой строки. Используя текстовые функции, можно объединить несколько строк в одну или, наоборот, разделить одну текстовую строку на несколько строк. Например, формула =СЦЕПИТЬ(A1;A2) объединяет две текстовые строки, содержащиеся в ячейках A1 и A2, в одну.

## Функции категории Проверка свойств и значений

Функции этой категории часто называют *информационными*. Функция ЯЧЕЙКА этой категории позволяет получить информацию о ячейке. Другие информационные функции проверяют выполнение какого-либо условия и, в зависимости от результата, возвращают значение ИСТИНА или ЛОЖЬ (или числовое значение). Например, с помощью функции ЕЧИСЛО можно проверить, данные какого типа содержит ячейка. Если в ячейке содержится число, функция ЕЧИСЛО возвращает логическое значение ИСТИНА, в противном случае функция возвращает логическое значение ЛОЖЬ.

## Функции Дата и время

Функции, принадлежащие к этой категории, предназначены для работы со значениями даты и времени. По сути, эти функции работают с числовыми значениями, потому что дата и время в Excel являются числами, к которым применен один из числовых форматов даты и времени. С помощью функции этой категории можно вычислить количество рабочих дней между двумя датами (функция ЧИСТРАБДНИ), преобразовать дату в год (функция ГОД), месяц (функция МЕСЯЦ) или день недели (функция ДЕНЬНЕД) и т.п.

## Математические функции

Математические функции позволяют выполнять простые и сложные вычисления. В категорию **Математические** входят тригонометрические функции, например SIN, COS, ACOS; функции, выполняющие арифметические действия, например СУММ, ПРОИЗВЕД, ЧАСТНОЕ; и многие другие функции. К этой же категории относятся функции, позволяющие работать с массивами значений или матрицами, — МУМНОЖ, МОПРЕД и МОБР, а также функции АГРЕГАТ и ПРОМЕЖУТОЧНЫЕ. ИТОГИ, которые используются для получения итоговых значений (суммы, среднего арифметического, минимального или максимального значений и т.п.) в массивах данных или списках.



Функция АГРЕГАТ доступна только в Excel 2010!

## Функции категории Ссылки и массивы

Эти функции (другое название этой категории — функции просмотра) используются для поиска значений в списках или таблицах (например, функции ПРОСМОТР, ВПР или ГПР), а также для получения дополнительной информации (например, функции ЧСТРОК, ЧИСЛСТОЛБ и ОБЛАСТИ возвращают количество строк, столбцов или областей в ссылке на диапазон). Функция ПОЛУЧИТЬ.ДААННЫЕ.СВОДНОЙ.ТАБЛИЦЫ извлекает данные, хранящиеся в сводной таблице. Некоторые функции этой категории позволяют манипулировать диапазоном данных. Например, функция ТРАНСП транспонирует диапазон ячеек на рабочем листе.

## Статистические функции и категория Совместимость

Статистические функции предназначены для проведения статистического анализа. С помощью статистических функций можно вычислить статистические характеристики набора данных, такие как среднее арифметическое значение (функция СРЗНАЧ), стандартное отклонение (функция СТАНДОТКЛОН.Г или СТАНДОТКЛОН.В) или дисперсия (функции ДИСП.В и ДИСП.Г), построить статистический ряд (функция ЧАСТОТА), вычислить доверительные интервалы (функции ДОВЕРИТ.НОРМ и ДОВЕРИТ.СТЬЮДЕНТ). В категории **Статистические** имеются также функции для проверки статистических гипотез, вычисления известных статистических распределений и проведения корреляционного и регрессионного анализа.

В версии Excel 2010 изменились алгоритмы и названия некоторых статистических функций. Например, функция СЛЧИС использует новый алгоритм получения случайных чисел; название этой функции не изменилось. Для вычисления дисперсии вместо функций ДИСП и ДИСПР используются функции ДИСП.В и ДИСП.Г соответственно. Функции ДИСП и ДИСПР остались для совместимости с предыдущими версиями и находятся в категории **Совместимость**. В Excel 2010 для вычисления дисперсии можно использовать как функции ДИСП.В и ДИСП.Г, так и функции ДИСП и ДИСПР.

## Инженерные функции

Инженерные функции используются, главным образом, в инженерных и научных расчетах. К инженерным функциям относятся:

- функции для работы с комплексными числами;
- функции, выполняющие преобразование чисел из одной системы счисления в другую (десятичную, восьмеричную, шестнадцатеричную и двоичную);
- функции, выполняющие преобразование величин из одной системы мер и весов в другую.

## Финансовые функции

Финансовые функции используются для решения финансовых и экономических задач. С помощью функций этой категории можно рассчитать приведенную стоимость (функция ПС), будущую стоимость (функция БС), оценить эффективность вложе-

ний при фиксированной процентной ставке, вычислить внутреннюю норму доходности и др. Анализ зависимости денежных сумм от времени, выполняемый с помощью финансовых функций, позволяет более обоснованно принимать решения экономического характера.

## Функции для работы с базами данных

Функции этой категории используются для анализа данных, содержащихся в обычных электронных таблицах (списках данных) и *таблицах* Excel. С помощью функций баз данных можно быстро получить нужную информацию, например подсчитать количество записей в базе данных (функция БСЧЁТ) или вычислить сумму значений (функция БДСУММ), удовлетворяющих некоторому условию. На ленте команд эта категория функций не отображается.

## Аналитические функции и куб данных OLAP

В традиционных базах данных типа OLTP (Online Transaction Processing – оперативная обработка транзакций), например Access или SQL Server, данные хранятся в нескольких таблицах, связанных между собой; каждая ячейка таблицы содержит один элемент данных. В Excel можно создать подключение к такой базе данных и получать оперативные данные из нее, например, через сводные таблицы. Однако, если в базе данных содержится несколько миллионов (или миллиардов) записей, Excel не в состоянии их обработать. В этом случае используют более эффективные средства обработки информации, например программу Microsoft SQL Server 2000, с помощью которой можно быстро обрабатывать большие объемы данных. Эту же программу можно использовать для создания хранилища данных и кубов OLAP.

Аббревиатура OLAP расшифровывается как Online Analytical Processing – Оперативный анализ данных. Сервер OLAP содержит средства, позволяющие связываться с хранилищем данных. Также сервер OLAP обрабатывает записи из хранилища данных и преобразует таблицы в куб данных. Таким образом, куб данных представляет собой многомерную структуру данных. В отличие от таблиц, используемых в реляционных базах данных, каждая ячейка куба содержит множество элементов. Для работы с кубом данных необходимо подключиться к нему и создать сводную таблицу.

Аналитические функции предназначены для работы с кубом данных OLAP. Например, функция КУБЭЛЕМЕНТ возвращает элемент или кортеж из куба, функция КУБСВОЙСТВОЭЛЕМЕНТА возвращает значение свойства элемента из куба.

## Пользовательские функции

К пользовательским функциям относятся все функции, которые были созданы пользователем. При вставке пользовательских функций в нижней части диалоговых окон **Аргументы функции** и **Мастер функций – шаг 1 из 2** не отображаются краткая справка о пользовательской функции и краткое описание аргументов этой функции. Это единственное отличие пользовательских функций от встроенных. В остальном с пользовательскими функциями можно работать так же, как и с любой встроенной функцией.

## Функции надстроек и программирования объектов

В Excel имеется еще одна категория функций — Внешние функции. Внешние функции загружаются с программами надстройки и обычно используются для получения каких-либо данных из внешних источников. К этой категории, например, относятся функции, преобразующие значение из национальных валют в евро и наоборот. Преобразование валют и форматирование данных в единицах евро становится доступным в Excel 2010 после установки надстройки “Инструменты для евро”. На ленте команд эти функции отображаются на вкладке **Формулы** в группе команд **Решения (Solutions)** после установки надстройки.

## Общие правила работы с функциями

Ввести функцию в формулу можно несколькими способами, но, независимо от способа ввода функций, необходимо


- правильно ввести имя функции;
- правильно задать аргументы функции.

Для многих функций также важен порядок следования аргументов. Для таких функций необходимо ввести аргументы в правильной последовательности.


## Ввод функций вручную

Ввести функцию в формулу можно вручную, и во многих случаях этот способ оказывается наиболее эффективным.

В Excel 2010 при вводе с клавиатуры любого имени открывается динамический раскрывающийся список доступных функций и имен, начинающихся с таких же букв, что и вводимый текст. На рис. 6.11 показан пример изменения раскрывающегося списка при вводе функции СУММ. После ввода первой буквы “с” в раскрывающемся списке отображается перечень всех доступных функций, начинающихся с буквы “с” (рис. 6.11, а). После ввода буквы “у” в списке останется 8 функций; первой в списке будет функция СУММ (рис. 6.11, б). Далее вы можете дважды щелкнуть на функции СУММ в раскрывающемся списке либо продолжить ввод функции, не обращая внимания на раскрывающийся список и то, что в нем отображается. После ввода имени функции этот список автоматически скроется.



а



б

Рис. 6.11. При вводе функции появляется динамический раскрывающийся список, содержащий список функций и имен. Вид списка после введения символов “с” (а) и “су” (б)



Если при вводе имени функции динамический список не появляется нажмите комбинацию клавиш <Alt+↓>.

После ввода имени функции и открывающей скобки, рядом с ячейкой или строкой формул появляется подсказка<sup>1</sup> (рис. 6.12), в которой отображается полный синтаксис функции: название функции и список ее аргументов, причем тот аргумент, который вводится в данный момент, выделен полужирным шрифтом.




Рис. 6.12. После ввода имени функции и открывающей скобки отображается подсказка, содержащая список аргументов функции

Вводить имя функции можно как строчными, так и прописными буквами. Если имя функции введено правильно, Excel автоматически преобразует символы имени функции в верхний регистр. Поэтому рекомендуется вводить имя функции строчными буквами. Например, если вы ввели формулы =сумм(A1:B4) или =срзнач(A1:B4), то по окончании ввода этих формул Excel преобразует символы имени функции в верхний регистр, т.е. в строке формул будут отображены формулы =СУММ(A1:B4) и =СРЗНАЧ(A1:B4) соответственно. Если программа не преобразовала имя функции в верхний регистр, это означает, что в названии функции допущена ошибка либо данная функция недоступна. Встроенные функции доступны из любой рабочей книги. Пользовательские функции доступны только в том случае, когда открыта рабочая книга (или загружена надстройка), содержащая программный код этой функции.



Если подсказка с названием и аргументами функции не появляется, откройте диалоговое окно Параметры Excel (команда Файл⇒Параметры) и на вкладке Дополнительно в группе Экран установите флажок Показывать всплывающие подсказки для функций.

<sup>1</sup> Такая же подсказка появляется в Excel 2002, 2003 и 2007, но отсутствует в более ранних версиях.

Рассмотрим процесс ввода функции на конкретном примере. Предположим, в ячейку D5 требуется ввести формулу =СУММ(A1:B4;D1:D4). Для этого необходимо выполнить следующие действия.

1. Введите знак равенства и имя функции строчными буквами (рис. 6.13, а). В процессе ввода имени функции можно использовать динамически раскрывающийся список.
2. Введите открывающую скобку.
3. Задайте аргументы функции:
  - выделите диапазон ячеек A1:B4;
  - введите точку с запятой, затем выделите диапазон D1:D4 (рис. 6.13, б);
 или
  - выделите диапазон A1:B4;
  - нажмите клавишу <Ctrl> и, удерживая ее нажатой, выделите диапазон D1:D4.
4. Введите закрывающую скобку и нажмите клавишу <Enter>.

Результат показан на рис. 6.13, в.




Рис. 6.13. Ввод функции вручную

## Вставка функции в формулу

Функцию можно вставить в формулу без использования клавиатуры. Для этого выполните одно из следующих действий.

- ▶ Щелкните на кнопке **Вставка функции**, которая находится в строке формул.
- ▶ Выберите команду **Формулы** ⇒ **Вставить функцию**.
- ▶ Нажмите комбинацию клавиш <Shift+F3>.

После выполнения любого из перечисленных выше действий откроется диалоговое окно **Мастер функций – шаг 1 из 2** (рис. 6.14), в котором содержится список категорий функций (раскрывающийся список **Категория**) и список доступных функ-



ций (список **Выберите функцию**). Для того чтобы выбрать функцию, выполните одно из действий.

- ▶ Выберите в раскрывающемся списке **Категория** нужную категорию, а затем в списке **Выберите функцию** — ту функцию, которую собираетесь вставить в формулу.
- ▶ Если не помните, к какой категории относится функция, выберите в раскрывающемся списке **Категория** опцию **Полный алфавитный перечень**. В списке **Выберите функцию** в алфавитном порядке будет выведен полный список функций. Выберите в этом списке нужную функцию.
- ▶ Чтобы выбрать функцию, которая использовалась одной из последних, выберите в раскрывающемся списке **Категория** опцию **10 недавно использовавшихся**. В списке **Выберите функцию** будет отображен список из 10 функций, которые использовались последними.

После выделения функции в нижней части диалогового окна **Мастер функций – шаг 1 из 2** отображается краткое ее описание.





Рис. 6.14. Диалоговое окно **Мастер функций – шаг 1 из 2**



Excel запоминает 10 функций, которые использовались последними. Если какая-либо функция используется часто, вы можете не открывать диалоговое окно **Мастер функций – шаг 1 из 2**. Активизируйте ячейку и введите знак равенства. В поле **Имя** в строке формул отобразится имя функции Excel. Щелкните на кнопке с треугольником рядом с названием этой функции, и откроется список из 10 функций, использовавшихся последними (рис. 6.15). Если нужной функции в этом списке не оказалось, выберите значение **Другие функции**, чтобы открыть диалоговое окно **Мастер функций – шаг 1 из 2**.

После выбора функции в диалоговом окне **Мастер функций – шаг 1 из 2** щелкните на кнопке **OK**. Откроется диалоговое окно **Аргументы функции**. Внешний вид этого диалогового окна определяется количеством аргументов функции. Например, если функция не имеет аргументов, в диалоговом окне **Аргументы функции** не будет ни одного текстового поля. Если функция имеет аргументы, то каждому аргументу

функции в диалоговом окне **Аргументы функции** соответствует конкретное поле. Название поля указано слева и совпадает с названием аргумента. На рис. 6.16 представлено диалоговое окно **Аргументы функции** для функции ЕСЛИ.



В диалоговом окне **Аргументы функции** можно определить аргументы функции. Посмотрим, как это сделать для функции ЕСЛИ. Предположим, на рабочий лист, показанный на рис. 6.17, в ячейку D6 необходимо ввести формулу =ЕСЛИ(A1<A2;D3;B4). Для этого выполните следующие действия.

Рис. 6.15. Выбор часто используемой функции из списка в поле **Имя**




Рис. 6.16. Диалоговое окно **Аргументы функции**

	A	B	C	D	E	F	G
1	25	15	50	10			
2	40	20	50	15			
3	80	25	50	20			
4	100	30	50	25			
6				405	=СУММ(A1:B4;D1:D4)		

Рис. 6.17. Исходный лист с данными

- Щелкните на кнопке **Вставить функцию**, расположенной в строке формул (или выберите команду **Формулы** ⇒ **Вставить функцию**). Откроется диалоговое окно **Мастер функций** – шаг 1 из 2.
- В раскрывающемся списке **Категория** выберите категорию **Логические**. Затем в списке **Выберите функцию** выберите функцию **ЕСЛИ** и щелкните на кнопке **ОК**. Откроется диалоговое окно **Аргументы функции** (рис. 6.16). Возле каждого поля слева указано название аргумента, справа от каждого поля указан тип аргумента — **ссылка, число, строка, логическое** или **любое**.

3. Щелкните на поле **Лог\_выражение**. Обратите внимание: в нижней части окна появится краткое описание этого аргумента. Щелкните на ячейке **A1** (или введите ее название вручную), затем введите знак "<" и щелкните на ячейке **A2** (или введите название этой ячейки вручную). Если вы все сделали правильно, то в поле **Лог\_выражение** у вас появится логическое выражение **A1<A2**, а справа от поля появится результат — логическое значение **ИСТИНА**.
4. Щелкните на поле **Значение\_если\_истина**, затем — на ячейке **D3**. Справа от поля появится значение, которое содержится в этой ячейке.
5. Щелкните на поле **Значение\_если\_ложь**, затем — на ячейке **B4**. Справа от поля появится значение, которое содержится в данной ячейке.
6. В диалоговом окне **Аргументы функции** отображаются значения аргументов и результат, вычисляемый функцией.
  - Если аргумент задан правильно, то справа от поля аргумента отображается значение этого аргумента. Ниже под списком аргументов отображается значение функции, как показано на рис. 6.18.
  - Если один из аргументов задан неправильно, рядом с полем для этого аргумента отображается значение ошибки (рис. 6.19). Очевидно, если один из аргументов функции задан неправильно, то функция также возвращает значение ошибки. На рис. 6.19 неправильно задан аргумент **Лог\_выражение**, Excel не распознает имя первого операнда — **A1A2** в логическом выражении **A1A2<A3**. Поэтому справа от этого поля отображается значение ошибки **#ИМЯ?**. Если не исправить ошибку, в ячейке с формулой также будет отображаться значение ошибки **#ИМЯ?**.
7. Щелкните на кнопке **ОК**, чтобы вставить функцию в формулу. Если вы передумали вводить функцию, щелкните на кнопке **Закреть** или на кнопке **Отмена** либо нажмите клавишу <Esc>.




Рис. 6.18. Все аргументы функции ЕСЛИ заданы правильно



Диалоговое окно **Аргументы функции** может частично или полностью закрывать рабочее поле. Вы можете временно свернуть его, щелкнув на одной из кнопок свертывания окна.




Рис. 6.19. Аргумент Лог\_выражение функции ЕСЛИ задан неправильно

## Вставка функции в существующую формулу

Диалоговые окна **Мастер функций – шаг 1 из 2** и **Аргументы функции** можно использовать для вставки функции в уже существующую формулу. Выделите ячейку с формулой и поместите курсор в том месте формулы, куда требуется вставить функцию. Затем откройте диалоговое окно **Мастер функций – шаг 1 из 2**, выберите нужную функцию и в диалоговом окне **Аргументы функции** определите аргументы выбранной вами функции.

## Изменение функции в формуле

Диалоговое окно **Аргументы функции** можно использовать также и в том случае, если вам нужно изменить аргументы функции, вставленной в формулу. Чтобы изменить функцию в диалоговом окне **Аргументы функции**, выполните следующие действия.

1. Активизируйте ячейку с формулой.
2. Установите курсор перед функцией, аргументы которой необходимо переопределить.
3. Щелкните на кнопке **Вставить функцию** в строке формул (или выберите команду **Формулы** ⇒ **Вставить функцию**). Откроется диалоговое окно **Аргументы функции**, в котором будет отображена данная функция с аргументами.
4. Внесите все необходимые изменения.
5. Щелкните на кнопке **ОК**, чтобы сохранить изменения. Если вы передумали изменять функцию, щелкните на кнопке **Отмена** или на кнопке **Закрыть** либо нажмите клавишу <Esc>.

## Использование ленты команд для вставки функции

Для вставки функции в формулу можно использовать одну из команд, расположенных на вкладке **Формулы** в группе **Библиотека функций**. Например, чтобы вставить функцию **ЕСЛИ**, выберите команду **Формулы** ⇒ **Логические** ⇒ **ЕСЛИ**. Для вставки функции **ПС**, вычисляющей приведенную к текущему моменту стоимость инве-

стиций, выберите команду **Формулы**⇒**Финансовые**⇒**ПС**. После выбора функции откроется диалоговое окно **Аргументы функции**. Чтобы вставить функцию из категорий **Инженерные**, **Аналитические**, **Статистические**, **Проверка свойств и значений** и **Совместимость**, выберите команду **Другие функции** (рис. 6.20).




Рис. 6.20. Команды группы Библиотека функций расположены на вкладке **Формулы**



Последней командой в списке функций, раскрываемом после выбора любой из перечисленных выше команд, является команда **Вставить функцию**. Выберите эту команду, чтобы открыть диалоговое окно **Мастер функций** — шаг 1 из 2.

## Получение справки по функциям

При работе с функциями необязательно помнить синтаксис каждой функции и тем более особенности использования каждой из них. Намного важнее уметь пользоваться подсказками, которые Excel отображает на каждом шагу при работе с функциями. Если функция вводится вручную, возле формулы появляется подсказка с полным списком аргументов функции. Если вставка функции выполняется через диалоговые окна **Мастер функций** – шаг 1 из 2 и **Аргументы функции**, то получить сведения о функции и ее аргументах можно непосредственно в этих диалоговых окнах. В диалоговых окнах **Мастер функций** – шаг 1 из 2 и **Аргументы функции** описание функции отображается внизу окна. В диалоговом окне **Аргументы функции** описание аргумента функции отображается после щелчка на поле, в которое вводится значение аргумента.

Если краткой справки оказывается недостаточно, можно обратиться за дополнительной информацией к справочной системе. В нижнем левом углу диалоговых окон **Мастер функций** – шаг 1 из 2 и **Аргументы функции** имеется ссылка **Справка по этой функции** (в более ранних версиях Excel вместо ссылки имеется кнопка вызова справочной системы). После щелчка на этой ссылке (или кнопке) откроется окно справочной системы с информацией о данной функции (рис. 6.21).




Рис. 6.21. В справочной системе Excel 2010 можно получить исчерпывающую информацию о встроенной функции

Для того чтобы получить дополнительную информацию о функции, необязательно открывать диалоговые окна **Мастер функций – шаг 1 из 2** или **Аргументы функции**. Щелкните на кнопке **Справка: Microsoft Excel**, которая отображается в строке с вкладками над лентой команд, или нажмите клавишу <F1>. В окне справочной системы найдите раздел “Справочник по функциям”. Щелкните на этом разделе, чтобы отобразить его содержимое. После этого щелкните на категории, а затем на функции. В правой части окна справочной системы отобразится описание функции (см. рис. 6.21). В каждой статье содержатся описание функции, примеры ее использования и причины появления ошибок, связанные с неправильным заданием аргументов функции.

Получить справку по функции можно непосредственно в диалоговом окне **Мастер функций – шаг 1 из 2**, в котором имеется поле **Поиск функции**. Введите в это поле одно или несколько ключевых слов, чтобы было понятно, что нужно вычислить, и щелкните на кнопке **Найти**. После этого в поле **Выберите функцию** отобразится

список функций, удовлетворяющих условиям поиска. Если ни одной функции не было найдено, в поле **Поиск функции** отобразится фраза **Пожалуйста, перефразируйте свой вопрос**. Внимательно относитесь к подбору ключевых слов. Логика поиска функций по ключевым словам в Excel 2010 отличается от той, что была в предыдущих версиях Excel. При поиске по ключевым словам “среднее значение” в списке найденных функций отображается всего одна функция — МЕДИАНА (рис. 6.22); в списке не окажется функций СРЗНАЧ, СРЗНАЧА, ДСРЗНАЧ, СРГЕОМ и т.п., которые действительно вычисляют средние значения для наборов данных.

## Ошибки, возникающие при работе с функциями

Перед тем как приступить к рассмотрению встроенных функций Excel, хотелось бы кратко остановиться на проблемах, с которыми сталкиваются как начинающие, так и опытные пользователи. Как правило, трудности при работе с функциями вызваны следующими причинами:

- ▶ синтаксическими ошибками;
- ▶ ошибками в имени функции (или в именах аргументов);
- ▶ неправильным использованием аргументов функции.

Если вы допустили синтаксическую ошибку, например после ввода аргументов функции забыли ввести закрывающую скобку, Excel автоматически добавит закрывающую скобку. Например, формула `=СУММ(A1:A10` будет автоматически исправлена на `=СУММ(A1:A10)`; при этом даже не появится сообщение об ошибке. Однако если формула более сложная, то вариант исправления формулы, предложенный Excel, может привести к ошибке вычислений. Например, если вместо формулы `=СУММ(A1:A10)+4` вы ввели формулу `=СУММ(A1:A10+4`, Excel автоматически исправит формулу следующим образом: `=СУММ(A1:A10+4)`; при этом сообщение об ошибке не появится. Исправленная формула вернет значение 84, т.е. вместо суммирования всех чисел в диапазоне A1:A10 будет использовано только одно значение (по умолчанию это то значение, которое находится в одной строке с формулой (рис. 6.23)). В подобных случаях следует еще раз внимательно проверить формулу и исправить ошибку.



Довольно часто подобные “автоматические исправления” проходят незаметно для пользователя, что впоследствии приводит к труднораспознаваемым ошибкам. Представим, что в формуле `=СРЗНАЧ(СУММ(C2:C7);СУММ(D2:D7))` пропущена одна из скобок, например введена формула `=СРЗНАЧ(СУММ(C2:C7;СУММ(D2:D7))`. Excel предложит добавить закрывающую скобку в конец формулы. В итоге получится совершенно другая формула

`=СРЗНАЧ(СУММ(C2:C7;СУММ(D2:D7)))`

Если результат этой формулы используется в дальнейших вычислениях, то выявить ошибку бывает очень непросто.




Рис. 6.22. Результат поиска по ключевым словам “среднее значение” в диалоговом окне **Мастер функций – шаг 1 из 2**

=СРЗНАЧ(СУММ(С2:С7);СУММ(D2:D7))																	
	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	
1	25	15	50	10				Примеры формул							В этой формуле вычисляется сумма чисел 4 и 80. Из всего диапазона A1:A10 используется только число 80, которое находится в той же строке, что и формула =СУММ(A1:A10+4)		
2	40	20	50	15				Правильные									
3	80	25	50	20				249 =СУММ(A1:A10)+4						Неправильные			
4	100	30	50	25										84 =СУММ(A1:A10+4)			
5				405 =СУММ(A1:B4;D1:D4)										104			
6				20 =ЕСЛИ(A1<A2;D3;B4)										4			
7														4			
8																	
9								6,87298 =КОРЕНЬ(B1)+3						#ЧИСЛО! =КОРЕНЬ(-B1)+3			
10																	
11																	
12																	
13																	
14																	
15								317,5						635			

Рис. 6.23. Синтаксические ошибки, допущенные при вводе функций, приводят к ошибкам в вычислениях либо к появлению значений ошибок

Если вы допустили ошибку в имени функции, Excel не сможет распознать имя функции или аргумента, и после нажатия клавиши <Enter> в ячейке будет отображаться значение ошибки #ИМЯ?. Такая же ошибка появится и в том случае, если допущена ошибка в имени аргумента.

Неправильное использование аргументов функции приводит к тому, что в ячейке с формулой отображается одно из значений ошибки.

- ▶ Ошибка #ЧИСЛО! отображается в том случае, если вместо числового аргумента используется неприемлемый аргумент. Например, аргументом функции КОРЕНЬ может быть только неотрицательное число. Если аргументом этой функции является отрицательное число, то функция КОРЕНЬ и соответственно формула =КОРЕНЬ(-B1)+3 возвращают значение ошибки #ЧИСЛО! (см. рис. 6.23).
- ▶ Ошибка #ЗНАЧ! отображается тогда, когда используется недопустимый тип аргумента. В примере, приведенном на рис. 6.24, в формуле =ЛЕВСИМВ(A2;-3)&"&B2 используется функция ЛЕВСИМВ, которая возвращает заданное количество первых символов в текстовой строке. Количество возвращаемых символов определяется вторым аргументом этой функции. Очевидно, что использование отрицательного числа для задания количества возвращаемых символов лишено всякого смысла. Поэтому функция ЛЕВСИМВ и соответственно формула =ЛЕВСИМВ(A2;-3)&"&B2 возвращают значение ошибки #ЗНАЧ!. Аналогично не имеет смысла знак "минус", стоящий перед первым аргументом функции ЛЕВСИМВ в ячейке D9.

=ЛЕВСИМВ(-A2;3)&"&B2							
	A	B	C	D	E	F	G
1	Данные						
2	Январь	2010					
3	Февраль	2010					
4							
5	Правильные формулы:			Янв 2010	=ЛЕВСИМВ(A2;3)&"&B2		
6				Фев 2010	=ЛЕВСИМВ(A3;3)&"&B3		
7							
8	Неправильные формулы:			#ЗНАЧ!	=ЛЕВСИМВ(A2;-3)&"&B2		
9				#ЗНАЧ!	=ЛЕВСИМВ(-A2;3)&"&B2		
10							

Рис. 6.24. Ошибка #ЗНАЧ! появилась за счет использования недопустимого типа аргумента в функции ЛЕВСИМВ



Так же как и при работе с обычными формулами, не содержащими функций, для того чтобы выяснить причину появления ошибки, можно воспользоваться стандартными инструментами проверки формул: отобразить зависимости между ячейками, выполнить проверку формул, отслеживать значения, отображаемые в окне контрольного значения. Для проверки более сложных формул, содержащих несколько функций, можно открыть диалоговое окно **Вычисление формулы**. Пошаговое отслеживание вычислений в формуле позволяет быстро определить, на каком именно шаге, т.е. при вычислении какой из вложенных функций возникает ошибка.

## Функции СУММ, СРЗНАЧ, МИН, МАКС, СЧЁТ и СЧЁТЗ

Функции СРЗНАЧ, МАКС, МИН, СЧЁТ, СЧЁТЗ предназначены для работы с наборами значений, содержащихся в некотором диапазоне ячеек. В диалоговом окне **Мастер функций – шаг 1 из 2** эти функции можно найти в категории **Статистические**. Функция СУММ относится к математическим и предназначена для подсчета суммы в некотором диапазоне ячеек. Несмотря на то что эти функции относятся к разным категориям и вычисляют различные параметры для некоторого набора числовых значений, между ними много общего. Они имеют похожий синтаксис и часто используются в вычислениях.

### Функция СУММ

Функция СУММ, вне всяких сомнений, является наиболее используемой функцией Excel. Функция СУММ имеет следующий синтаксис:

**СУММ(число1;число2; ...;число255)**

где **число1**, **число2** – количество аргументов (от 1 до 255), сумму которых необходимо вычислить.

Аргументами функции СУММ должны быть числа, заданные в виде конкретных значений, ссылок на ячейки или диапазоны ячеек, массивов констант. Вместо ссылок можно использовать имена ячеек или диапазонов ячеек. Массив констант представляет собой массив чисел, заключенных в фигурные скобки, например {1;2;3} или {1:2:3}. Числа в фигурных скобках должны разделяться точкой с запятой либо двоеточием. Предположим, что в ячейках A1, A2 и A3 содержатся значения 1, 2 и 3. Тогда формулы =СУММ(1;2;3), =СУММ(A1:A2:A3), =СУММ(A1:A3), =СУММ(Данные) и =СУММ({1;2;3}) возвращают один и тот же результат – 6. Здесь имя **Данные** присвоено диапазону A1:A3. В первой формуле в качестве аргументов функции СУММ используются ссылки на ячейки, во второй формуле аргументом функции СУММ является массив значений.

С помощью функции СУММ можно подсчитать следующее.

- ▶ Сумму чисел, содержащихся в одном диапазоне ячеек. В этом случае достаточно задать один аргумент (**число1**). Для указания ссылки на непрерывный диапазон ячеек используется двоеточие (:) в качестве разделителя между первой и последней ячейками диапазона. Например, на рис. 6.25 формула =СУММ(C5:E5) в ячейке F5 возвращает сумму чисел, содержащихся в диапазоне ячеек C5:E5.

- ▶ Сумму чисел, содержащихся в нескольких диапазонах (как смежных, так и несмежных). В этом случае можно задать до 255 аргументов. Для создания ссылки на два несмежных диапазона используется оператор объединения диапазонов, обозначаемый точкой с запятой (;). Например, формула `=СУММ(C5:C7;C9:C11;C13:C15;C17:C19)` в ячейке C24 (рис. 6.25) возвращает сумму чисел, которые находятся в диапазонах C5:C7, C9:C11, C13:C15 и C17:C19.
- ▶ Сумму чисел, содержащихся в диапазоне, который является пересечением диапазонов, заданных в качестве аргументов. Для создания ссылки на такой диапазон используется оператор пересечения диапазонов – пробел. Например, формула в ячейке C21

`=СУММ(C$5:C$7 C$5:$E5;C$9:C$11 C$9:$E9;C$13:C$15 C$13:$E13;C$17:C$19 C$17:$E17)`

возвращает сумму чисел, которые содержатся на пересечении диапазонов: C5:C7 и C5:E5 (ячейка C5), C9:C11 и C9:E9 (ячейка C9), C13:C15 и C13:E13 (ячейка C13), C17:C19 и C17:E17 (ячейка C17), т.е. заданная таким образом функция вычисляет сумму чисел, содержащихся в ячейках C5, C9, C13 и C17 (см. рис. 6.25).

В последней формуле использованы смешанные ссылки на диапазоны. Использование смешанных ссылок позволяет сократить время на ввод достаточно громоздких формул в диапазон C21:E23. Достаточно ввести только одну формулу в ячейку C21 и затем скопировать ее в остальные ячейки диапазона C21:E23.

Если на рабочем листе определены имена, то использование имен в качестве аргументов функции СУММ делает формулы если не менее громоздкими, то, по крайней мере, более осмысленными (рис. 6.26).

		C21					
		=СУММ(C\$5:C\$7 C\$5:\$E5;C\$9:C\$11 C\$9:\$E9;C\$13:C\$15 C\$13:\$E13;C\$17:C\$19 C\$17:\$E17)					
	A	B	C	D	E	F	G
		Название магазина	Легковые автомобили	Грузовые автомобили	Микроавтобусы	Всего	Формулы
4							
5	Квартал I	Альянс	720 000	453 200	890 000	2 063 200	=СУММ(C5:E5)
6		Автосалон	220 600	215 000	325 000	760 600	=СУММ(C6:E6)
7		ЛВ-Авто	379 100	798 200	650 000	1 827 300	=СУММ(C7:E7)
8		Итого	1 319 700	1 466 400	1 865 000	4 651 100	=СУММ(F5:F7)
9	Квартал II	Альянс	420 000	400 500	620 200	1 440 700	=СУММ(C9:E9)
10		Автосалон	280 900	495 200	650 200	1 426 300	=СУММ(C10:E10)
11		ЛВ-Авто	340 200	560 100	360 700	1 261 000	=СУММ(C11:E11)
12	Итого	1 041 100	1 455 800	1 631 100	4 128 000	=СУММ(F9:F11)	
13	Квартал III	Альянс	290 300	345 500	560 000	1 195 800	=СУММ(C13:E13)
14		Автосалон	512 300	548 200	445 200	1 505 700	=СУММ(C14:E14)
15		ЛВ-Авто	500 200	450 100	350 900	1 301 200	=СУММ(C15:E15)
16	Итого	1 302 800	1 343 800	1 356 100	4 002 700	=СУММ(F13:F15)	
17	Квартал IV	Альянс	350 800	325 400	810 700	1 486 900	=СУММ(C17:E17)
18		Автосалон	780 200	789 500	405 100	1 974 800	=СУММ(C18:E18)
19		ЛВ-Авто	609 500	250 300	412 300	1 272 100	=СУММ(C19:E19)
20	Итого	1 740 500	1 365 200	1 628 100	4 733 800	=СУММ(F17:F19)	
21	За год	Альянс	<b>1 781 100</b>	1 524 600	2 880 900	6 186 600	=СУММ(C21:E21)
22		Автосалон	1 794 000	2 047 900	1 825 500	5 667 400	=СУММ(C22:E22)
23		ЛВ-Авто	1 829 000	2 058 700	1 773 900	5 661 600	=СУММ(C23:E23)
24		Итого	5 404 100	5 631 200	6 480 300	17 515 600	=СУММ(C24:E24)
25							=СУММ(E5:E7;E9:E11;E13:E15;E17:E19)
26							=СУММ(D5:D7;D9:D11;D13:D15;D17:D19)
27							=СУММ(C5:C7;C9:C11;C13:C15;C17:C19)

Рис. 6.25. В функции СУММ можно использовать оператор пересечения диапазонов

	A	B	C	D	E	F	G
	Название магазина	Легковые автомобили	Грузовые автомобили	Микроавтобусы	Всего	Формулы	
4	Альянс	720 000	453 200	890 000	2 063 200	=СУММ(Альянс_1_кв)	
5	Автосалон	220 600	215 000	325 000	760 600	=СУММ(Автосалон_1_кв)	
6	ЛВ-Авто	379 100	798 200	650 000	1 827 300	=СУММ(ЛВ_Авто_1_кв)	
7	Итого	1 319 700	1 466 400	1 865 000	4 651 100	=СУММ(ЛВ_Авто_1_кв;Автосалон_1_кв;Альянс_1_кв)	
8	Альянс	420 000	400 500	620 200	1 440 700	=СУММ(Альянс_2_кв)	
9	Автосалон	280 900	495 200	650 200	1 426 300	=СУММ(Автосалон_2_кв)	
10	ЛВ-Авто	340 200	560 100	360 700	1 261 000	=СУММ(ЛВ_Авто_2_кв)	
11	Итого	1 041 100	1 455 800	1 631 100	4 128 000	=СУММ(Альянс_2_кв;Автосалон_2_кв;ЛВ_Авто_2_кв)	
12	Альянс	290 300	345 500	560 000	1 195 800	=СУММ(Альянс_3_кв)	
13	Автосалон	512 300	548 200	445 200	1 505 700	=СУММ(Автосалон_3_кв)	
14	ЛВ-Авто	500 200	450 100	350 900	1 301 200	=СУММ(ЛВ_Авто_3_кв)	
15	Итого	1 302 800	1 343 800	1 356 100	4 002 700	=СУММ(Альянс_3_кв;Автосалон_3_кв;ЛВ_Авто_3_кв)	
16	Альянс	350 800	325 400	810 700	1 486 900	=СУММ(Альянс_4_кв)	
17	Автосалон	780 200	789 500	405 100	1 974 800	=СУММ(Автосалон_4_кв)	
18	ЛВ-Авто	609 500	250 300	412 300	1 272 100	=СУММ(ЛВ_Авто_4_кв)	
19	Итого	1 740 500	1 365 200	1 628 100	4 733 800	=СУММ(Альянс_4_кв;Автосалон_4_кв;ЛВ_Авто_4_кв)	
20	Альянс	1 781 100	1 524 800	2 880 900	6 186 600	=СУММ(Альянс_год)	
21	Автосалон	1 794 000	2 047 900	1 825 500	5 667 400	=СУММ(Автосалон_год)	
22	ЛВ-Авто	1 829 000	2 058 700	1 773 900	5 661 600	=СУММ(ЛВ_Авто_год)	
23	Итого	5 404 100	5 631 200	6 480 300	17 515 600	=СУММ(Альянс_год;Автосалон_год;ЛВ_Авто_год)	
24				=СУММ(Микроавтобусы)			
25				=СУММ(Грузовые_автомобили)			
26							
27				=СУММ(Легковые_автомобили)			

Рис. 6.26. Тот же пример, что и на рис. 6.25, но в качестве аргументов функции СУММ используются имена диапазонов

## Особенности использования функции СУММ

При суммировании значений с помощью функции СУММ учитываются только числовые значения, представленные в одном из стандартных числовых форматов Excel, а также числа, представленные как текст. Аргументы, которые ссылаются на пустые ячейки, логические или текстовые значения, игнорируются (рис. 6.27).

	A	B	C	D	E	F	G	H	I	J
1	Диапазон 1				Диапазон 2					
2	1	11	21		1	11	ИСТИНА			логическое значение
3	2	12	22		2	12	22			
4	3	13	23		3	13	23			
5	4	14	24		4	14	текст			текст
6	5	15	25		ИСТИНА		25			
7	15	65	115		8	50	70			число, представленное как текст
8										
9										
10										
11										
12										
13										
14										
15										
16										
17										
18										
19										

Рис. 6.27. Примеры использования функции СУММ

Отметим, что логическое значение ИСТИНА может влиять на результат вычислений функции СУММ. Например, формула =СУММ(1;3;4;ИСТИНА), в которой логическое значение ИСТИНА указано явно, вернет значение 9. Формула =СУММ

({1;3;4;ИСТИНА}), в которой в качестве аргумента функции СУММ используется массив констант, возвращает тот же результат, что и формула =СУММ(E2:E6).

Если логические значения непосредственно используются в качестве аргументов функции СУММ, то они преобразуются в числа. Логическое значение ИСТИНА преобразуется в число 1, а значение ЛОЖЬ — в 0. Вот почему формула =СУММ(1;3;4;ИСТИНА) (см. рис. 6.27) возвращает другой результат по сравнению с формулой =СУММ(E2:E6).

Если хотя бы один из аргументов функции СУММ является значением ошибки, функция СУММ возвращает значение ошибки (рис. 6.28).

	A	B	C	D	E	F	G
1	Диапазон 1				Диапазон 2		
2	1	11	21		#ИМЯ?	11	ИСТИНА
3	2	12	22		2	#Н/Д	22
4	3	13	23		3	#ИМЯ?	23
5	4	14	24		4	14	текст
6	5	15	25		#Н/Д	15	#ЧИСЛО!
7	15	65	115		#ИМЯ?	#Н/Д	#ЧИСЛО!
8	=СУММ(A2:A6)	=СУММ(B2:B6)	=СУММ(C2:C6)		=СУММ(E2:E6)	=СУММ(F2:F6)	=СУММ(G2:G6)
9							

Рис. 6.28. Функция СУММ возвращает значение ошибки, если ее аргументы ссылаются на значение ошибки

## Функции СЧЁТ и СЧЁТЗ

Функция СЧЁТ подсчитывает количество ячеек, содержащих числовые значения. Так же как и функция СУММ, функция СЧЁТ может иметь до 255 аргументов. Аргументами могут быть числа, ссылки на ячейки и диапазоны или имена ячеек и диапазонов, массивы констант. Ячейки могут содержать данные разных типов, однако в подсчете участвуют только ячейки с числовыми значениями или числами, представленными как текст. Пустые ячейки, а также ячейки, содержащие значение ошибки, логические значения или текст, игнорируются (рис. 6.29). Если в ячейке содержится дата, представленная в одном из стандартных числовых форматов, то эта ячейка также учитывается при подсчете.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Диапазон 1				Диапазон 2					Диапазон 3			
2	1	11	21		1	11.фев	21			1	11.фев	21	
3	2	12	22	текст		12	22		текст		12	22	
4	0	13	23		ЛОЖЬ	13	23				13	23	
5	4	0	1		4	0	ИСТИНА			4	0	ИСТИНА	
6	5	15	25		5	15	25			5	15	25	
7	12	51	92	СУММ	10	40260	91	СУММ		10	40260	91	СУММ
8	5	5	5	СЧЕТ	3	5	4	СЧЕТ		3	5	4	СЧЕТ
9	5	5	5	СЧЕТЗ	5	5	5	СЧЕТЗ		4	5	5	СЧЕТЗ
10													

Рис. 6.29. Подсчет ячеек с помощью функций СЧЁТ и СЧЁТЗ

Функция СЧЁТЗ, так же как и функция СЧЁТ, используется для подсчета количества ячеек с данными в некотором диапазоне ячеек, но в отличие от функции СЧЁТ, учитывающей при подсчете только ячейки с числовыми значениями, функция СЧЁТЗ подсчитывает количество непустых ячеек. Если ячейка содержит символ

пробела (т.е. значение " ") или формулу, которая возвращает пустую строку, то она считается непустой и учитывается при подсчете. Если аргументом является массив значений или ссылка, то пустые ячейки в массиве или ссылке игнорируются. На рис. 6.29 показано, как функция СЧЁТЗ выполняет подсчет непустых ячеек.



Ячейка, содержащая символ пробела или формулу, которая возвращает пустую строку (т.е. значение " "), не считается пустой. Это необходимо учитывать при подсчете количества ячеек с помощью функции СЧЁТЗ. На рис. 6.30 ячейки A16 и F16 внешне ничем не отличаются. Однако из этих двух ячеек пустой является только ячейка F16. Ячейка A16 содержит формулу, возвращающую пустую строку. Кроме того, функция СЧЁТЗ учитывает при подсчете также и те ячейки, в которых содержится значение ошибки.

	A	B	C	D	E	F	G	H	I	J
13	Диапазон 1					Диапазон 2				
14	1	11.фев	21			1	11.фев	21		
15	2	12	22			текст	12	22		
16		13	23				#Н/Д	23		
17	текст	0	ИСТИНА			4	0	ИСТИНА		
18	5	15	25			5	15	25		
19	8	40260	91	СУММ		10	#Н/Д	91	СУММ	
20	3	5	4	СЧЕТ		3	4	4	СЧЕТ	
21	5	5	5	СЧЕТЗ		4	5	5	СЧЕТЗ	
22	Ячейки, не содержащие числа:					6 =СЧЁТЗ(A14:C18;F14:H18)-СЧЁТ(A14:C18;F14:H18)				
23										

Рис. 6.30. Функция СЧЁТЗ подсчитывает количество непустых ячеек

Особенности работы функций СЧЁТ и СЧЁТЗ можно использовать при подсчете ячеек, содержащих нечисловые значения. Например, формула в ячейке E22 на рис. 6.30 возвращает значение 6, т.е. всего шесть ячеек в диапазонах A14:C18 и F14:H18 содержат нечисловые значения.

## Функция СРЗНАЧ

Функция СРЗНАЧ вычисляет среднее арифметическое своих аргументов. Если заданы  $n$  действительных чисел  $a_1, a_2, \dots, a_n$ , то число

$$A_n = \frac{a_1 + a_2 + \dots + a_n}{n}$$

называется *средним арифметическим* чисел  $a_1, a_2, \dots, a_n$ .

В формуле для среднего арифметического сначала вычисляется сумма  $n$  чисел, затем полученный результат делится на количество слагаемых. Для вычисления среднего арифметического  $n$  чисел в Excel можно использовать одну из следующих формул:

- ▶ =СРЗНАЧ(Диапазон1)
- ▶ =СУММ(Диапазон1)/СЧЁТ(Диапазон1)

Пример использования этих формул показан на рис. 6.31. В качестве аргументов функций СРЗНАЧ, СУММ и СЧЁТ используются имена Диапазон1 и Диапазон2, содержащие ссылку на диапазоны A2:C6 и E2:G6 соответственно. Всего функция СРЗНАЧ может иметь до 255 аргументов.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	Диапазон 1				Диапазон 2				Диапазон 3_1					
2	1	11	21		1	11	21		1	11	21			
3	2	12	22		2	12	22		2	12	22			
4	0	13	23		0	13	текст		0	13	0			
5	0	14	24		ИСТИНА	14	24		4	14	24			
6	5	15	25		5	15	25		5	15	25			
7	8	65	115	СУММ	8	65	92	СУММ	12	65	92	СУММ		
8	5	5	5	СЧЕТ	4	5	4	СЧЕТ	5	5	5	СЧЕТ		
9	5	5	5	СЧЕТЗ	5	5	5	СЧЕТЗ	5	5	5	СЧЕТЗ		
10	1,6	13	23	СРЗНАЧ	2	13	23	СРЗНАЧ	2,4	13	18,4	СРЗНАЧ		
11	=СУММ(Диапазон1)				=СУММ(Диапазон2)				=СУММ(Диапазон3_1)					
12	=СЧЕТ(Диапазон1)				=СЧЕТ(Диапазон2)				=СЧЕТ(Диапазон3_1)					
13	=СУММ(Диапазон1)/СЧЕТ(Диапазон1)				=СУММ(Диапазон2)/СЧЕТ(Диапазон2)				=СУММ(Диапазон3_1)/СЧЕТ(Диапазон3_1)					
14	=СРЗНАЧ(Диапазон1)				=СРЗНАЧ(Диапазон2)				=СРЗНАЧ(Диапазон3_1)					
15	=СРЗНАЧА(Диапазон1)				=СРЗНАЧА(Диапазон2)				=СРЗНАЧА(Диапазон3_1)					
16	=СУММ(Диапазон1)/СЧЕТЗ(Диапазон1)				=СУММ(Диапазон2)/СЧЕТЗ(Диапазон2)				=СУММ(Диапазон3_1)/СЧЕТЗ(Диапазон3_1)					
17														

Рис. 6.31. Использование функции СРЗНАЧ для вычисления среднего арифметического значения

Как видно из рис. 6.31, при вычислении среднего арифметического функция СРЗНАЧ игнорирует логические и текстовые значения, но учитывает нулевые значения.



Если при определении среднего арифметического необходимо учитывать пустые ячейки, логические значения и текст, используйте функцию СРЗНАЧА. При вычислении среднего арифметического с помощью функции СРЗНАЧА логическое значение ИСТИНА полагается равным 1, логическое значение ЛОЖЬ и текст (кроме чисел, представленных в виде текста) — равными 0 (нулю). На рис. 6.31 хорошо видно, что функции СРЗНАЧ и СРЗНАЧА возвращают разные значения для одного и того же диапазона данных. Кроме того, если исходный диапазон содержит логическое значение ИСТИНА, формулы =СРЗНАЧА(Диапазон2) и =СУММ(Диапазон2)/СЧЕТЗ(Диапазон2) возвращают разные значения.

## Функции МАКС и МИН

Функция МАКС возвращает наибольшее значение из набора значений, функция МИН — наименьшее значение. В качестве аргументов обеих функций могут использоваться числа, ссылки на ячейки или диапазоны ячеек, имена ячеек или диапазонов, массивы констант. Аргументов может быть до 255.

При определении максимального и минимального значений с помощью функций МАКС и МИН учитываются только те ячейки, которые содержат числа; пустые строки, текст (за исключением чисел, представленных в виде текста) и логические значения игнорируются (рис. 6.32).

Функции МИН и МАКС возвращают значение ошибки, если хотя бы одна ячейка диапазона содержит значение ошибки. Если диапазон не содержит ячеек с числовыми значениями, функции МАКС и МИН возвращают значение 0 (нуль).



Если при определении наибольшего и наименьшего значений в заданном диапазоне требуется учитывать пустые ячейки, логические значения и текст, используйте функции МАКСА и МИНА соответственно. На рис. 6.33 хорошо видно, что функции МИН и МИНА, МАКС и МАКСА возвращают разные значения для одного и того же диапазона.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1		Набор значений 1 (имя Данные1)				Набор значений 2 (имя Данные2)				Набор значений 3 (имя Данные3)				
2		1	11	21		1	11				1	11		21
3		2	12	22		2	12	22			2	12		22
4		0	13	23			13	текст				13		текст
5		4	14	24		4	14	24			4	14		#Н/Д
6		5	15	25		5	15	ЛОЖЬ			5	15		ИСТИНА
7		12	65	115	СУММ	12	65	67	СУММ		12	65	#Н/Д	СУММ
8		5	5	5	СЧЕТ	4	5	3	СЧЕТ		4	5	2	СЧЕТ
9		5	5	5	СЧЕТЗ	4	5	5	СЧЕТЗ		4	5	5	СЧЕТЗ
10		0	11	21	МИН	1	11	21	МИН		1	11	#Н/Д	МИН
11		5	15	25	МАКС	5	15	24	МАКС		5	15	#Н/Д	МАКС
12		0 =МИН(Данные1)				1 =МИН(Данные2)				#Н/Д =МИН(Данные3)				
14		25 =МАКС(Данные1)				24 =МАКС(Данные2)				#Н/Д =МАКС(Данные3)				

Рис. 6.32. Примеры использования функций МАКС и МИН

	A	B	C	D	E	F	G	H	I
16		Набор значений 4				Набор значений 5			
17						11.фев	5	21	
18		текст				2	12	22	
19				ИСТИНА		3	13	ЛОЖЬ	
20						4	14	24	
21		0 =МИН(B17:D20)				2 =МИН(F17:H20)			
22		0 =МАКС(B17:D20)				40220 =МАКС(F17:H20)			
23		0 =МИНА(B17:D20)				0 =МИНА(F17:H20)			
24		1 =МАКСА(B17:D20)				40220 =МАКСА(F17:H20)			
25		диапазон содержит дату							

Рис. 6.33. Функции МИН и МИНА, МАКС и МАКСА возвращают разные значения

## Команды Сумма и Автосумма

Функции СУММ, МИН, МАКС, СРЗНАЧ, СЧЁТ и СЧЁТЗ используются в вычислениях довольно часто. В Excel 2010 для быстрой вставки этих функций в формулу на ленте имеются две команды. Команда **Сумма** расположена на вкладке Главная в группе Редактирование. Команда **Автосумма** находится на вкладке Формулы в группе Библиотека функций.

Для того чтобы вставить в ячейку функцию СУММ, активизируйте ячейку, в которую необходимо ввести формулу, и выберите команду **Формулы** ⇒ **Автосумма** либо команду **Главная** ⇒ **Редактирование** ⇒ **Сумма**. В ячейке появится знак равенства и имя функции СУММ с подсказкой, в которой отображается список аргументов функции СУММ. Excel автоматически выделит один из диапазонов с числами, если текущая ячейка смежная с этим диапазоном, — вокруг выделенного диапазона возникнет движущаяся рамка, в текущей ячейке этот диапазон будет выделен (рис. 6.34). Если нужно указать другой диапазон, выделите его с помощью мыши. Вы увидите, что в формуле вместо исходной ссылки появилась ссылка на другой диапазон. Если необходимо задать несколько аргументов функции СУММ, при выделении несмежных диапазонов используйте

	A	B	C	D
1	25	15	5	
2	40	30	20	
3	55	45	35	
4	70	60	50	
5				
6				
7				

Формула в ячейке: =СУММ(C1:C4)

Рис. 6.34. Щелкните на кнопке **Автосумма**: Excel вставит в формулу функцию СУММ и автоматически выделит диапазон, смежный с числовыми значениями

клавишу <Ctrl>. При выделении следующего диапазона Excel автоматически вставит точку с запятой в список аргументов, а затем добавит новую ссылку.



Для вставки функции СУММ в формулу используйте комбинацию клавиш <Alt+=>.

Для того чтобы вставить в формулу другие функции — СРЗНАЧ, МИН, МАКС и СЧЁТ, — необходимо щелкнуть на маленькой кнопке с черным треугольником, расположенной рядом с кнопкой **Автосумма** или **Сумма**. Щелчок на этой кнопке открывает меню из шести команд (рис. 6.35).

- ▶ **Сумма.** При выборе этой команды Excel автоматически вставляет в текущую ячейку функцию СУММ, как было описано выше.  
Вставка остальных функций выполняется аналогично.
- ▶ **Среднее.** Выполняется вставка функции СРЗНАЧ.
- ▶ **Число.** Выполняется вставка функции СЧЁТ.
- ▶ **Максимум.** Выполняется вставка функции МАКС.
- ▶ **Минимум.** Выполняется вставка функции МИН.
- ▶ **Другие функции.** Эта команда открывает диалоговое окно **Мастер функций** — шаг 1 из 2, в котором содержится список доступных функций.




Рис. 6.35. Меню команды **Автосумма**

## Вычисления в строке состояния

В Excel можно выполнять простейшие вычисления, не вводя формулы на рабочий лист.




Мы рекомендуем использовать этот способ в тех случаях, когда рабочий лист заполнен данными и для выполнения каких-либо промежуточных расчетов нет времени либо места на рабочем листе.



Если щелкнуть правой кнопкой мыши на строке состояния, появится контекстное меню (рис. 6.36), в котором можно установить флажки возле следующих значений.

- ▶ **Среднее.** Вычисляется среднее значение в выделенном диапазоне. Аналог функции СРЗНАЧ.
- ▶ **Количество.** Вычисляется количество ячеек, содержащих любые данные в выделенном диапазоне. Аналог функции СЧЁТЗ.
- ▶ **Количество чисел.** Вычисляется количество ячеек, содержащих числовые данные, в выделенном диапазоне. Аналог функции СЧЁТ.
- ▶ **Минимум.** Вычисляется минимальное числовое значение из выделенного диапазона. Аналог функции МИН.
- ▶ **Максимум.** Вычисляется максимальное числовое значение из выделенного диапазона. Аналог функции МАКС.
- ▶ **Сумма.** Вычисляется сумма чисел в выделенном диапазоне. Аналог функции СУММ.

Контекстное меню строки состояния



Установите флажки возле этих команд, чтобы включить режим автоматических вычислений

Рис. 6.36. Установите флажки возле соответствующих команд контекстного меню строки состояния, чтобы при выделении диапазона ячеек выполнялись вычисления

Для того чтобы включить режим автоматических вычислений, выполните следующие действия.

1. Выделите диапазон ячеек.

- Щелкните правой кнопкой мыши на строке состояния. Откроется контекстное меню.
- Установите флажки возле соответствующих команд контекстного меню.

Теперь при выделении диапазона ячеек в строке состояния будет отображаться результат автоматических вычислений. Например, если в контекстном меню выбрана команда **Количество**, то при выделении диапазона Excel автоматически подсчитает количество значений в этом диапазоне и отобразит в строке состояния результат: **Количество=12**.

Для того чтобы отключить режим автоматических вычислений, откройте контекстное меню строки состояния и снимите флажки возле соответствующих команд.

## Резюме

Данная глава является введением в функции Excel. В этой главе вы узнали, что собой представляет встроенная функция Excel, как правильно ввести функцию в формулу и получить подробную справку о функции. Вы научились также работать с функциями СУММ, СРЗНАЧ, СРЗНАЧА, МИН, МИНА, МАКС, МАКСА, СЧЁТ и СЧЁТЗ.

В следующей главе рассматриваются логические, текстовые и информационные функции.

## Вопросы и ответы

**После ввода формулы =СРЗНАЧ(В1:В12) на пустой рабочий лист в ячейке с формулой отображается значение ошибки #ДЕЛ/0!. Как устранить эту ошибку?**

Среднее значение в диапазоне В1:В12 можно вычислить с помощью двух других функций — СУММ и СЧЁТ, а именно  $\text{СРЗНАЧ(В1:В12)=СУММ(В1:В12)/СЧЁТ(В1:В12)}$ . Из этой формулы становится понятно следующее. Если количество ячеек, содержащих числа, равно нулю либо все ячейки являются пустыми, мы получаем деление на нуль. Функция СРЗНАЧ возвращает значение ошибки #ДЕЛ/0!, если ее аргумент (или все ее аргументы) ссылается на пустые ячейки. Для того чтобы эта ошибка не отображалась на пустом рабочем листе, вместо формулы =СРЗНАЧ(В1:В12) введите формулу =ЕСЛИ(СЧЁТ(В1:В12)=0;"";СРЗНАЧ(В1:В12)). Функция ЕСЛИ выполняет проверку условия — СЧЁТ(В1:В12)=0 и, если данное условие неверно, вычисляет среднее арифметическое всех чисел в указанном диапазоне, в противном случае возвращает пустую строку.

**Мне необходимо вычислить сумму трех чисел. Какую формулу для этого лучше использовать: =СУММ(А2:А4) или =А2+А3+А4?**

Когда требуется вычислить сумму двух–пяти значений, как в вашем случае, преимущества использования функции СУММ не настолько очевидны, как, например, при суммировании 10 и больше чисел. На первый взгляд, обе формулы кажутся равноправными. Все же мы советуем использовать формулу =СУММ(А2:А4). При изменении исходного диапазона (добавлении новых строк в диапазон или удалении имеющихся строк из диапазона) Excel автоматически изменит ссылку на исходный диапазон в первой формуле (см. пример, представленный на рис 6.37).

	A	B	C		A	B	C		A	B	C
1	Исходный диапазон			1	После вставки новой строки			1	После удаления строки 3		
2	10			2	10			2	10		
3	20			3	15			3	30		
4	30			4	20			4	40	#ССЫЛКА!	
5	60	60		5	30			5	=СУММ(A2:A3)	=A2+ССЫЛКА!+A3	
6	=СУММ(A2:A4)	=A2+A3+A4		6	75	60		6	удалена строка 3		
7				7	=СУММ(A2:A5)	=A2+A4+A5		7	из исходного диапазона		
8				8				8			
9				9	новая строка			9			
10				10				10			

Рис. 6.37. Преимущество использования функций в формулах при изменении исходного диапазона

